

Массивы

Понятие массива

Массив – это набор элементов одинакового типа, имеющих общее имя. Доступ к конкретному элементу массива осуществляется через указание в скобках после имени его индекса.

Массив характеризуется размером и размерностью.

Размер – это количество элементов в массиве, **размерность** – количество индексов, необходимых для указания местоположения элемента в массиве.

Таким образом, массив представляет собой совокупность

од



ннх

$i=1$	8	7	2	12
$i=2$	5	1	0	3
$i=3$	8	7	9	15
	$j=1$	$j=2$	$j=3$	$j=4$

i - строки
 j - столбцы

Типы массивов

Язык VBA поддерживает два типа массивов – статические и динамические.

Статическими называют такие массивы, размер которых был указан непосредственно при их объявлении. В этом случае размер массива остается фиксированным на протяжении всего выполнения программы.

Динамические массивы имеют переменное количество элементов, т.е. динамические массивы могут увеличиваться или сокращаться, в зависимости от того, какое количество элементов требуется в данный момент.

Объявление массивов

Статические массивы.

Прежде чем использовать массив, нужно обязательно объявить его с помощью оператора **Dim** и указать тип хранящихся в массиве значений. Все значения в массиве принадлежат к одному типу данных.

Синтаксис оператора объявления массива следующий:

```
Dim имяМассива (размер1, размер2, ...) As типДанных
```

где указанные в скобках величины *размер1, размер2* задают размеры массива – количество индексов и максимально допустимое значение для каждого конкретного индекса. При этом индексирование элементов массива по умолчанию начинается с нуля.

```
Dim имяМассива (нижняя граница To верхняя граница, ...) As типДанных
```

При объявлении массива можно указать не только верхнюю границу индекса, но и его нижнюю границу, т. е. явно задать диапазон изменения конкретного индекса массива, причем нижняя граница может быть любым целым числом, не обязательно неотрицательным

Примеры объявления статических массивов

```
Dim A(9) As Integer
```

определяет одномерный массив с именем A из 10 элементов, являющихся переменными целого типа

```
Dim B(4, 9) As Variant
```

определяет двумерный массив 5×10 из 50 элементов, являющихся переменными типа Variant.

```
Dim B(1 To 3, 1 To 3) As Single
```

```
Dim Z(1 To 11) As Integer
```

В качестве стандартного значения нижней границы массива (индекса) может использоваться не только ноль. Чтобы изменить это стандартное значение, нужно воспользоваться оператором `Option Base`.

Если поместить в начало модуля оператор `Option Base 1`, то индексирование элементов массивов по умолчанию будет начинаться не с нуля, а с единицы.

Например, в следующем операторе объявляется одномерный массив `Z`, состоящий из 11 элементов.

```
Option Base 1
```

```
Dim Z(11) As Integer
```

Объявление массивов

Динамические массивы.

Синтаксис объявления и определения размеров динамического массива:

1. Объявляется массив с использованием оператора `Dim`, но без указания размерности. Признаком массива являются скобки после его имени

```
Dim имяМассива() As типДанных
```

2. Перед использованием массива в программе описывается данный массив с указанием размерности при помощи оператора `ReDim`.

```
ReDim имяМассива(размер1, размер2, ...)
```

где указанные в скобках величины `размер1`, `размер2` задают размеры массива – количество индексов и максимально допустимое значение для каждого конкретного индекса.

При этом индексирование элементов массива по умолчанию начинается с нуля, если не используется оператор `Option Base 1`.

Ввод элементов одномерного массива

Вводить элементы массива можно разными способами.

- С помощью оператора присваивания, если массив небольшого размера.

Option Base 1

```
Dim Z(4) As Integer
```

```
Z(1) = 2 : Z(2) = 7 : Z(3) = 3 : Z(4) = 9
```

- В цикле, если элементы массива задаются формулой или используется генератор случайных чисел.

Пример заполнения статического массива A(12), элементы которого вычисляются по формуле.

```
Dim A(12) As Single, i As Integer
```

```
For i = 1 To 12
```

```
A(i) = i ^ 2
```

```
Next i
```

- Пример заполнения динамического массива Z(), элементы которого задаются генератором случайных чисел.

```
Dim A(12) As Integer, i As Integer
```

```
For i = 1 To 12
```

```
A(i) = Int(RND*100-50)
```

```
Next i
```


- Заполнение массива при помощи функции *InputBox()*

```
Dim A(12) As Single, i As Integer
For i = 1 To 12
A(i) = InputBox ("Введите " & Str(i) & " элемент массива", _ "Ввод данных")
Next i
```

- Заполнение массива данными с рабочего листа

```
Dim A(8) As Single, i As Integer
For i = 1 To 8
A(i) = Cells(1, i).Value
Next i
```

	A	B	C	D	E	F	G	H	I
1	16	23	45	8	34	25	87	32	
2									

```
Dim A(8) As Single, i As Integer
For i = 1 To 8
A(i) = Cells(i+3, 1).Value
Next i
```

	A	B	C	D
1				
2				
3				
4	16			
5	23			
6	45			
7	8			
8	34			
9	25			
10	87			
11	32			
12				
13				

Вывод одномерного массива

- Вывод массива на рабочий лист

For i = 1 To N

```
Cells(1, i).Value = Z(i)
```

Next i

- Вывод массива в список ListBox

```
Dim A(8) As Single, i As Integer
```

'Ввод массива

For i = 1 To 8

```
A(i) = Cells(1, i).Value
```

Next i

```
ListBox1.List = A ' Вывод массива
```


Другой способ

For i = 1 To 8


```
A(i) = Cells(1, i).Value 'Ввод массива
```

```
ListBox1.AddItem A(i) 'Вывод массива
```

Next i




- Вывод массива в текстовое поле



The screenshot shows the VBA code editor for 'Книга1 - UserForm2 (Code)'. The 'Declarations' tab is selected. The code defines an array A of 8 single-precision floating-point numbers and a string t. It then loops through the array, reading values from the first row of the worksheet into the array and concatenating them into the string t. Finally, it displays the string t in TextBox1.

```
Option Base 1
Private Sub CommandButton1_Click()
    Dim A(8) As Single, i As Integer, t As String
    t = ""
    For i = 1 To 8
        A(i) = Cells(1, i).Value 'Ввод массива
        t = t & Str(A(i)) & " "
    Next i
    TextBox1.Text = t
End Sub
```



The screenshot shows the runtime window for 'UserForm2'. It contains a label 'Исходный массив' (Initial array) above a text box displaying the values '16 23 45 8 34 25 87 32'. A button labeled 'Вычислить' (Calculate) is located at the bottom right.

Типовые алгоритмы

- Нахождение суммы, произведения, среднего значения
- Нахождение суммы, произведения, среднего значения, количества при выполнении условия
- Поиск (максимального, минимального и т.д.)
- Преобразование (сортировка, вставка, удаление и т.д.)

Типовые алгоритмы

Пример 1. Определить сумму, произведение, среднее значение в массиве $A(8)$

	A	B	C	D	E	F	G	H	I
1	6	2	4	8	7	5	8	3	
2									

UserForm2

Исходный массив

6 2 4 8 7 5 8 3

Сумма: 43

Произведение: 322560

Среднее значение: 5,375

Вычислить


Выход

```
Книга1 - UserForm2 (Code)
CommandButton1 Click
Option Base 1
Private Sub CommandButton1_Click()
    Dim A(8) As Single, i As Integer, t As String
    Dim S As Single, P As Double, Sr As Single
    t = ""
    For i = 1 To 8
        A(i) = Cells(1, i).Value 'Ввод массива
        t = t & Str(A(i)) & " "
    Next i
    TextBox1.Text = t
    'Нахождение суммы и среднего значения
    S = 0
    For i = 1 To 8
        S = S + A(i)
    Next i
    TextBox2.Text = S
    Sr = S / 8
    TextBox4.Text = Sr
    'Нахождение произведения
    P = 1
    For i = 1 To 8
        P = P * A(i)
    Next i
    TextBox3.Text = P
End Sub

Private Sub CommandButton2_Click()
End
End Sub
```

Типовые алгоритмы

Пример 2. Определить среднее значение положительных элементов в массиве $Z(N)$




Ввод данных

Введите количество элементов

12

OK Cancel




UserForm3

20
3
7
-22
-20
27
-49
26
31
20
-46
-9

Среднее значение положительных элементов в массиве равно 16,6

Вычислить

Выход



```
Option Base 1
Private Sub CommandButton1_Click()
    Dim Z() As Single, i As Integer, N As Integer
    Dim S As Single, k As Integer, Sr As Single
    N = InputBox("Введите количество элементов", "Ввод данных")
    ReDim Z(N)
    For i = 1 To N
        Z(i) = Int(Rnd * 100 - 50)
        ListBox1.AddItem Z(i)
    Next i
    'Нахождение суммы и среднего значения положительных элементов
    S = 0: k = 0
    For i = 1 To N
        If Z(i) > 0 Then S = S + Z(i): k = k + 1
    Next i
    Sr = S / k
    Label1.Caption = "Среднее значение положительных элементов в массиве равно " & Sr
End Sub
```

Пример 3. Определить максимальное значение и его номер в массиве R(M)

Ввод данных

Введите количество элементов

OK Cancel

18

UserForm4

Максимальное значение 48

Порядковый номер 16

Вычислить Выход

44
-14
2
26
-45
9
-4
-21
12
14
-24
-23
32
32
8
48
41
-28

Книга1 - UserForm4 (Code)

CommandButton1 Click

```
Option Base 1

Private Sub CommandButton1_Click()
    Dim R() As Single, i As Integer, M As Integer
    Dim max As Single, nomer As Integer
    M = InputBox("Введите количество элементов", "Ввод данных")
    ReDim R(M)
    For i = 1 To M
        R(i) = Int(Rnd * 100 - 50)
    Next i
    ListBox1.List = R
    'Нахождение максимального значения и его номера
    max = R(1): nomer = 1
    For i = 1 To M
        If R(i) > max Then max = R(i): nomer = i
    Next i
    TextBox1.Text = max
    TextBox2.Text = nomer
End Sub
```



Пример 4. Выполнить сортировку по убыванию в массиве Z(N)

Метод пузырька

Идея – пузырек воздуха в стакане воды поднимается со дна вверх.

Для массивов – самый маленький («легкий» элемент перемещается вверх («всплывает»)).


1-ый проход




- начиная снизу, сравниваем два соседних элемента; если они стоят «неправильно», меняем их местами
- за 1 проход по массиву **один** элемент (самый маленький) становится на свое место



2-ой проход



3-ий проход



Для сортировки массива из N элементов нужен N-1 проход (достаточно поставить на свои места N-1 элементов).

Пример 4. Выполнить сортировку по убыванию в массиве Z(N)

Ввод данных

Введите N

OK Cancel

12

Сортировка

Массив	Сортировка
12	48
14	48
-24	41
-23	32
32	32
32	19
8	14
48	12
41	8
-28	-23
19	-24
48	-28

Вычислить

Очистить

Выход

```
Одномерные массивы.xlsm - UserForm6 (Code)
CommandButton2 Click

Private Sub CommandButton1_Click()
    ListBox1.Clear
    ListBox2.Clear
End Sub

Private Sub CommandButton2_Click()
    Dim Z() As Single, X() As Single
    Dim i As Integer, N As Integer, k As Integer
    N = InputBox("Введите N", "Ввод данных")
    ReDim Z(N), X(N)
    For i = 1 To N
        Z(i) = Int(Rnd * 100) - 50
        ListBox1.AddItem Z(i)
    Next i
    'Сортировка по убыванию
    For i = 1 To N - 1
        For k = i + 1 To N
            If Z(i) < Z(k) Then c = Z(i): Z(i) = Z(k): Z(k) = c
        Next k
    Next i
    For i = 1 To N
        ListBox2.AddItem Z(i)
    Next i
End Sub

Private Sub CommandButton3_Click()
End
End Sub
```

Пример 5. Имеются итоги выполнения месячного задания по товарообороту двенадцатью магазинами фирмы:

Определить

Номер магазина	1	2	3	4	5	6	7	8	9	10	11	12
Выполнение задания, %	102,1	100,2	97,5	110,2	99,7	103,3	101,9	94,8	108,6	105,0	103,7	99,4

- 1. значение и номер магазина с минимальным выполнением задания
- 2. Количество магазинов с выполнением задания ниже среднего

	A	B
1	Номер магазина	Выполнение задания
2	1	102.1
3	2	100.2
4	3	97.5
5	4	110.2
6	5	99.7
7	6	103.3
8	7	101.9
9	8	94.8
10	9	108.6
11	10	105.0
12	11	103.7
13	12	99.4
14		

Пример 4

Номер магазина	Выполнение задания	Минимальное значение	94,8
1	102.1	Номер магазина	8
2	100.2	Среднее значение	102,2
3	97.5	Количество магазинов с выполнением задания ниже среднего	7
4	110.2		
5	99.7		
6	103.3		
7	101.9		
8	94.8		
9	108.6		
10	105.0		
11	103.7		
12	99.4		

их номера 1 2 3 5 7 8 12

Вычислить

Выход

Одномерные массивы.xlsm - UserForm5 (Code)

CommandButton1 Click

```
Option Base 1
Private Sub CommandButton1_Click()
    Dim p(12) As Single, i As Integer
    Dim min As Single, nm As Integer
    Dim k As Integer, S As Single, sr As Single, t As String
    'заполнение ListBox1 с рабочего листа
    ListBox1.ColumnCount = 2 'количество столбцов в ListBox1
    ListBox1.RowSource = "A1:B13"
    'Ввод массива
    For i = 1 To 12
        p(i) = Val(Cells(i + 1, 2))
    Next i
    'Нахождение минимального значения и его номера
    min = p(1): nm = 1
    For i = 1 To 12
        If p(i) < min Then min = p(i): nm = i
    Next i
    TextBox1.Text = min
    TextBox2.Text = nm
    'Нахождение среднего значения
    S = 0
    For i = 1 To 12
        S = S + p(i)
    Next i
    sr = S / 12
    TextBox3.Text = sr
    'Количество магазинов с выполнением задания ниже среднего
    k = 0: t = ""
    For i = 1 To 12
        If p(i) < sr Then k = k + 1: t = t & Str(i) & " "
    Next i
    TextBox4.Text = k
    Label5.Caption = "их номера " & t
End Sub

Private Sub CommandButton2_Click()
End
End Sub
```

Пример 6. Выполнить преобразование в массиве Z(N), вычислить количество и выбрать только нечетные элементы

Преобразование массива

20	3	Количество нечетных элементов <input type="text" value="6"/>
3	7	
7	27	
-22	-49	
-20	31	
27	-9	
-49		
26		
31		
20		
-46		
-9		

Вычислить

Очистить

Выход

```
Одномерные массивы.xlsm - UserForm7 (Code)
CommandButton1 Click

Private Sub CommandButton1_Click()
    Dim Z() As Single, X() As Single
    Dim i As Integer, N As Integer, k As Integer
    N = InputBox("Введите N", "Ввод данных")
    ReDim Z(N), X(N)
    For i = 1 To N
        Z(i) = Int(Rnd * 100) - 50
        ListBox1.AddItem Z(i)
    Next i

    'Выбор нечетных
    k = 0
    For i = 1 To N
        If Z(i) Mod 2 <> 0 Then k = k + 1: X(k) = Z(i)
    Next i
    TextBox1.Text = k
    For i = 1 To k
        ListBox2.AddItem X(i)
    Next i
End Sub
```

Пример 7. В массиве Z(N) поменять местами первый и максимальный элементы.

UserForm8

Массив

36	46
29	29
-13	-13
46	36
37	37
-45	-45
44	44
-14	-14
2	2
26	26
-45	-45
9	9

Максимальное значение: 46

номер: 4

Вычислить

Выход

Одномерные массивы.xlsm - UserForm8 (Code)

CommandButton1 Click

```
Option Base 1
Private Sub CommandButton1_Click()
    Dim Z() As Single, i As Integer, N As Integer
    Dim max As Single, nomer As Integer, c As Single
    N = InputBox("Введите количество элементов N", "Ввод данных")
    ReDim Z(N)
    For i = 1 To N
        Z(i) = Int(Rnd * 100 - 50)
    Next i
    ListBox1.List = Z
    'Нахождение максимального значения и его номера
    max = Z(1): nomer = 1
    For i = 1 To N
        If Z(i) > max Then max = Z(i): nomer = i
    Next i
    TextBox1.Text = max
    TextBox2.Text = nomer
    'меняем местами первый и максимальный
    c = Z(1): Z(1) = Z(nomer): Z(nomer) = c
    'выводим преобразованный массив
    ListBox2.List = Z
End Sub
```